

BAYERISCHE AKADEMIE DER WISSENSCHAFTEN

MATHEMATISCH-NATURWISSENSCHAFTLICHE KLASSE

ABHANDLUNGEN · NEUE FOLGE, HEFT 143

NOVA KEPLERIANA

Neue Folge – Heft 3

Eine doppelte Iterationsrechnung von
Johannes Kepler und ihre Programmierung

Zu seiner Berechnung der scheinbaren Planetenbahn

Bearbeitet von

VOLKER BIALAS

Vorgelegt von Herrn Walther Gerlach
in der Sitzung vom 10. Januar 1969

MÜNCHEN 1970

VERLAG DER BAYERISCHEN AKADEMIE DER WISSENSCHAFTEN

IN KOMMISSION BEI DER C.H.BECK'SCHEN VERLAGSBUCHHANDLUNG MÜNCHEN

Druck der C. H. Beck'schen Buchdruckerei Nördlingen
Printed in Germany

INHALT

1. Historische Einführung	5
2. Das Rechenverfahren der Rudolphinischen Tafeln	7
innere Iteration	8
äußere Iteration	10
3. Rechenbeispiel	10
4. Beschreibung des ALGOL-Programms	14
5. Die Ergebnisse der Nachrechnung – kritischer Vergleich mit den Werten Keplers	19
die Merkurrechnungen in den Handschriften	23
6. Die Lösung des Problems über eine Differentialgleichung	25
7. Schlußbetrachtung	29
Zusammenfassung	29
Literaturnachweis	30

1. HISTORISCHE EINFÜHRUNG

Termini Stationum I O V I S.		
In Anomalia Eccentri.	Primæ.	Secundæ.
	Angulus Com-	mutationis.
0	123.57	124.54
90	126.22	126.24
180	128.15	127.15
270	125.38	125.41


Kepler: Tabulae Rudolphinae

Im Mittelpunkt des astronomischen Interesses stand seit der Antike bis in die Jahrhunderte der modernen Wissenschaft hinein der Lauf der Planeten. Beobachtungen und Rechnungen, praktische und theoretische Astronomie hatten gleiches Terrain für den Gegenstand ihrer Betrachtungen, bis Kepler mit dem grundsätzlich neuen Ansatz, Erkenntnisse der Physik zur Erforschung der Ursachen der Himmelsbewegungen heranzuziehen, der Theorie den neuen Weg in die Himmelsmechanik andeutete. Gleichwohl war die tatsächliche Wirkung auf die rechnende Astronomie zunächst gering.

Für die Leser astronomischer Bücher und Benutzer astronomischer Tafelwerke des 17. Jahrhunderts war auch weiterhin die Diskussion um das Weltsystem oder um die wahre Form der Bahn schwer durchschaubar oder uninteressant. Ihnen kam es mehr darauf an, die vorhergesagten Planetenörter in zufriedenstellender Übereinstimmung mit den tatsächlichen Örtern zu wissen.

In seinem astronomischen Tafelwerk, TABULAE RUDOLPHINAE [1], hat Johannes Kepler die Darstellung der heliozentrischen Planetenbewegung in elliptischen Bahnen zum Hauptgegenstand des Werkes gemacht. Daneben ist er in den Planetentafeln auch auf die scheinbare Planetenbewegung eingegangen in kleinen Tabellen jeweils am Schluß der Tafeln eines Planeten: TERMINI STATIONUM. Die verschlungene Bahn, für die alte Astronomie Veranlassung, nach dem aristotelischen Dogma der Kreisbewegung die Epizykeltheorie in die Astronomie einzuführen, – diese sichtbare Bahn der Planeten erklärte Kopernikus als Überlagerungseffekt der Planetenbewegung mit der Erdbewegung um die Sonne.

Kepler betrachtet die scheinbare Bahn in ausgezeichneten Punkten, in denen sich die Vorwärtsbewegung zur Rückwärtsbewegung umkehrt und die rückläufige Bewegung wieder rechtläufig wird.




Figur 1

Dabei kommt es südlich und nördlich der Ekliptik zu Schleifenbildungen, wenn der zweite rechläufige Zweig den vorher beschriebenen Zweig schneidet; die Kurve bildet einen Doppelpunkt. In der Nähe der Ekliptik ändert sich die Breite verhältnismäßig rasch, so daß die Schleifenbildung bei der Bewegungsumkehrung ausbleiben kann.

Kepler war nicht der erste, der die Umkehrpunkte der Planetenbewegung durch Vorausberechnung bestimmt hat, so originell auch sein Verfahren einer doppelten Iterationsrechnung sein mag. Vor ihm hatte Kopernikus die Ursachen der Schleifenbewegung erkannt und die stationären Punkte berechnet, aber schon Ptolemäus vermochte die Umkehrpunkte im voraus zu bestimmen.

Nach Kopernikus durchläuft etwa der Planet Jupiter (♃) den Bogen AB um die mittlere Sonne S, wenn die Erde einen vollen Umlauf von E nach E beschreibt. Die mit den Be-




Figur 2

wegungen von Erde und Jupiter vorwärts laufenden Sichtstrahlen Erde-Jupiter weisen für eine kurze Zeitspanne zur Zeit der ersten Station (STATIO PRIMA) an dieselbe Stelle der Himmelssphäre (Richtung 1-1'), sie weisen dann im rückläufigen Teil der Bahn immer weiter nach rückwärts, bis sie in der zweiten Station (STATIO SECUNDA) wieder für eine kurze Zeit in dieselbe Richtung weisen (Richtung 2-2'). Nach Passieren der Linie 2-2' laufen die Sichtlinien wieder vorwärts.

Die Stationen 1' und 2' begrenzen also den rückläufigen Bogen des Planetenlaufes und liegen symmetrisch zur Opposition SPO.

Die Rechnungen von Kopernikus („De Revolutionibus“, V 35 u. 36) gehen auf die von Ptolemäus im „Almagest“ wiedergegebenen Überlegungen zurück, nach denen Streckenverhältnisse mit Geschwindigkeitsverhältnissen verglichen werden.

Eine vom Mittelpunkt Z der Ekliptik gezogene Gerade schneidet den Epizykel (Exzenter) in Θ und A so, daß die Hälfte des innerhalb des Kreises gelegenen Abschnittes sich



Figur 3

zu dem Abschnitt außerhalb des Kreises verhält wie die Geschwindigkeit des Epizykels zur Geschwindigkeit des Planeten, also

$$E\Theta : \Theta Z = \widehat{CE} : \widehat{FA}.$$

Mit dieser Formulierung des Problems kann Ptolemäus einen Satz von Apollonius heranziehen, der den Schnitt der größten Seite eines Dreiecks durch eine Gerade behandelt und den Zusammenhang zwischen Winkeln und Seiten angibt.

Kopernikus folgt dieser Betrachtung und überträgt die mathematischen Beziehungen auf sein System, indem er etwa die Epizykelbewegung eines oberen Planeten durch die Bewegung der Erde um die Sonne ersetzt.

2. DAS RECHENVERFAHREN DER RUDOLPHINISCHEN TAFELN

In [1] geht Kepler von der Definition der stationären Stellung eines Planeten aus. Auf S. 72 schreibt er:

„Da ja zweifellos die Sichtlinie zu einem Planeten nach einem Tag der Länge nach zum selben Punkt des Tierkreises zeigt wie tags zuvor, sind die Sichtstrahlen entweder identisch – wie bei Tycho – oder parallel – wie bei Kopernikus...“

Kepler erläutert das Problem an einer Figur, die hier als Fig. 4 wiedergegeben wird (vgl. auch [2], Abschnitt 1.21).

In Fig. 4 seien:

S die Sonne

A ein äußerer Planet


O die Erde; oder

A die Erde und

O ein innerer Planet;

D der Ort von A nach einem Tag

P der Ort von O nach einem Tag.




Figur 4

OA und PD sind die parallelen Sichtstrahlen zur Zeit der Statio.

AE ist die Parallele zu SOI durch A, AB die Parallele zu OP. Von A wird auf PIC das Lot AF gefällt.

Einzelheiten über Keplers Überlegungen zum Rechengang in [1] sind aus [2] ersichtlich. Wichtig ist, daß die Winkel bei C, E, D, B, A als Zwischengrößen berechnet werden und daß sich die gesuchte Kommutation (Angulus Commutationis), die die stationäre Stellung des Planeten anzeigt, über den Komplementwinkel CAE ergibt.



Figur 5

Für die Lösung der Aufgabe nach Kepler sind weiter zwei Größen bedeutungsvoll, auf die in [2] ausführlich eingegangen wird: das Verhältnis der Abstände Sonne-Planet zu Sonne-Erde $\left(\frac{AC}{AE}\right)$ und das Verhältnis der Tagbögen $\left(\frac{AB}{AD}\right)$. Der Abstand Sonne-Planet wird, da die Kommotation ein Winkel in der Ekliptikebene ist, auf diese Ebene reduziert.

Mit der mittleren Bewegung m von Sonne und Planet berechnet sich das Verhältnis der Tagbögen über das dritte Keplersche Gesetz zu

$$\frac{AB}{AD} = \frac{\Delta M_2}{\Delta M_1} \sqrt{a_2}. \quad (1)$$

Index 1 bezeichnet die Erde, 2 den (äußeren) Planeten.

Der Tagbogen d ergibt sich nach

$$d = m \frac{\Delta E}{\Delta M}. \quad (2)$$

ΔE ist der Zuwachs (etwa 1°) der vorgegebenen oder durch die mittlere Anomalie M bestimmten exzentrischen Anomalie, ΔM der dem Intervall ΔE und der Anomalie entsprechende Zuwachs in M .

Die Berechnung der Kommotation läuft auf eine Iterationsrechnung hinaus, die im Zusammenhang mit der gesamten Rechnung als *innere Iteration* bezeichnet werden soll. Eine derartige Rechnung gliedert sich in mehrere Schritte (vgl. [2]):

1. Das Verhältnis der Tagbögen nach (1) wird $\text{cosec } \alpha$ gesetzt, mit $\alpha \approx D$, also

$$\frac{AB}{AD} = \text{cosec } D \text{ in erster Näherung;} \quad (3)$$

C ist sicher kleiner als $(90 - D)$,

$$C < 90 - D. \quad (4)$$

2. Da AF gleich 1 gesetzt ist und für ein beliebiges Dreieck mit der Kathete AF der Zusammenhang besteht:

Hypotenuse = cosec Gegenwinkel von AF ,
erhält man mit $D = 90 - C$ und dem bekannten Verhältnis von (1)

$$\text{cosec } B = \frac{AB}{AD} \text{ cosec } D. \quad (5)$$

3. Da nach dem Sehnentangentensatz

$$EAB = 90 + \frac{1}{2} \text{ OSP} \quad (6)$$

und
$$\text{CAD} = 90 + \frac{1}{2} \text{ASD}, \quad (7)$$

ergeben sich
$$E = 180 - (\text{EAB} + B) \quad (8)$$

und für einen verbesserten Wert von C

$$\text{cosec } C = \frac{AC}{AE} \text{ cosec } E. \quad (9)$$

4. Damit ist die erste Durchrechnung beendet, die Iteration geht vom verbesserten Winkel C aus. D berechnet sich jetzt genauer nach

$$D = 180 - (\text{CAD} + C). \quad (10)$$

Die Rechnung wird über die Gleichungen (5), (8), (9), (10) fortgesetzt und schließlich abgebrochen, wenn die Verbesserungen für C zu Null werden.

Nach beendeter Iteration wird die Kommotation berechnet:


$$A = 180 - (E - C). \quad (11)$$

In [1] bestimmt Kepler in einem Beispiel drei Werte für C und interpoliert nach der Regula falsi, bis die Verbesserung für C zu Null wird.

In den Tabellen der Stationspunkte in [1] hat Kepler die Kommotation für Statio prima und Statio secunda in Abhängigkeit von der exzentrischen Anomalie angegeben. Es könnte hiernach den Anschein haben, als wäre die Kommotation allein abhängig von der Stellung des Planeten in der Bahn.

Der Hinweis darauf, daß die Entfernung Sonne-Planet in die Ekliptikebene projiziert wird, zeigt die Abhängigkeit von Neigung und Länge des Knotens bei der Bildung des Arguments der Breite.

In den Rechenbeispielen von [1] findet sich kein Anhaltspunkt für eine Bezugsepoche, ein Umstand, der eine genaue Berechnung der Kommotation ebenso wenig zuließe wie die Unterscheidung der Stationen in primae und secundae. Die Unterscheidung der stationären Punkte, in Keplers *EPITOME ASTRONOMIAE COPERNICANAE* nur angedeutet, wird in [1], S. 73, 74 ausführlich behandelt.



Figur 6

Es wird der Zusammenhang zwischen Kommotation und Anomalie der Sonne über die Distanz des Planetenaphels vom Sonnenapogäum hergestellt. Die Beziehungen sind aus Fig. 6 leicht ersichtlich.

Die Übersetzung der betreffenden Stelle ([1], S. 73) lautet:
 „Die ermittelte Kommotation wird für die Statio prima zur wahren Anomalie addiert (für Statio secunda wird die Differenz Anomalie-Kommotation gebildet)... In beiden Fällen wird die Apheldistanz des Planeten vom Apogäum der Sonne hinzugefügt; die Summe ist die wahre Anomalie der Sonne, die Tagbogen und Intervall bestimmt. Mit diesen Daten werden beide Verhältnisse (der Entfernungen und der Tagbögen) korrigiert.“

Diese Stelle beinhaltet die Gleichung:

$$v_{\odot} = v \pm A + \text{Aphel}_{\text{Pl.}} - \text{Apogäum}_{\text{Sonne}} \quad (12)$$

+ gilt für Statio prima,
 — gilt für Statio secunda.

Die Ausführungen geben nicht nur die Anleitung zur Berechnung der Statio secunda, sie bieten vor allem die Möglichkeit der durchgreifenden Kontrolle, indem die neu bestimmte wahre Anomalie der Sonne mit der früheren Anomalie verglichen wird.

Es wird eine zweite Iterationsrechnung erforderlich, die *äußere Iteration*, die der inneren übergeordnet ist und von der Kommotation nach der inneren Iteration ausgeht. In diese Rechnungen nach (12) gehen Aphel des Planeten und Apogäum der Sonne ein, Richtungen, die ebenso wie die Knotenlinie von der Bezugsepoche abhängen. Angaben darüber lassen sich, wie schon gesagt, in [1] nicht finden.

Indessen bleiben Spekulationen in dieser Richtung erspart. In den erhaltenen Keplerhandschriften [3] befindet sich auch das Druckmanuskript zu den Rudolphinischen Tafeln mit Abschnitten, die beim Druck ausgelassen wurden. Unter diesen Passagen, über die im Zusammenhang mit Rechenbeispielen noch ausführlich zu sprechen sein wird, läßt sich die Bezugsepoche finden, die der Berechnung der Tafeln Termini Stationum zugrunde gelegt ist ([3], Bd. 20, Bl. 29 u. 29^V): ANNUS 1625 COMPLETUS.

Damit sind Rechenverfahren und Ausgangswerte Keplers bekannt. Es ergibt sich die günstige Situation, daß die Ergebnisse Keplers nachgerechnet werden können. Die Tabellen Termini Stationum in [1] enthalten für Saturn, Jupiter und Venus Kommotationen für jeweils vier exzentrische Anomalien (0, 90, 180, 270) und für Mars und Merkur Kommotationen für jeweils acht Anomalien (0, 60, 90, 120, 180, 240, 270, 300). Berücksichtigt man noch die Unterscheidung der Stationes, ergeben sich insgesamt 56 Kommotationen.

Um die Routinearbeit in der Nachrechnung zu vermeiden, wird ein ALGOL-Programm für die elektronische Rechenanlage TR4 des Leibniz-Rechenzentrums der Bayerischen Akademie der Wissenschaften formuliert. Ehe auf dieses Rechenprogramm eingegangen wird, sollen die Rechenschritte beider Iterationen an einem vollständigen Beispiel der Stationsrechnungen verdeutlicht werden.

3. RECHENBEISPIEL

Es sollen die Anguli Commutationis für Jupiter mit der exzentrischen Anomalie $E = 90^\circ$ berechnet werden. Bezugsepoche ist 1. Januar 1626, 12^h. Aus den Tafeln von [1] für Jupiter und Sonne sind die Ausgangswerte der Rechnung zu berechnen:

Jupiter: Aphel in $187^{\circ}11'40''$
 Knoten in $95.27.26$
 $E = 90$ $\Delta M = 60'$
 $v = 87.14.10$
 $r = 5.200.00$
 $d = 5'$ nach (2)
 wahrer Bahnort in $274^{\circ}25'50''$
 Knoten in $95.27.26$

 Argument der Breite $178.58.24$
 Verkürzung von $r = 0$.

Sonne: Da keine Angaben über die Größe der Kommotation vorausgesetzt werden und die stationären Punkte annähernd symmetrisch zur Opposition liegen, wird zunächst von der Opposition ausgegangen.

wahrer Ort in $94^{\circ}25'50''$
 Apogäum in $96.9.49$

 $v = 358.16.1$ $\Delta M = 61'5''$
 $r = 1.101799$
 $d = 58'$ nach (2).

Verhältnis der Tagbögen: $\frac{AB}{AD} = 2.23991$ nach (1)

Verhältnis der Abstände: $\frac{AC}{AE} = 5.10811$.

Die Rechnung folgt nun dem Rechenschema der inneren Iteration.

1. Durchrechnung: $D = 29^{\circ}57'$ nach (3)
 $\operatorname{cosec} B = 2.23991 \cdot 2.00303$
 $B = 12^{\circ}52'40''$ nach (5)
 $E = 76.38.20$ nach (8)
 $\operatorname{cosec} C = 5.10811 \cdot 1.02782$
 $C = 10^{\circ}58'50''$ nach (9)
 $D = 78.58.40$ nach (10).

Damit ist der Ausgangswert verbessert und die erste Durchrechnung beendet. Es folgen die Iterationen.

1. Iteration: $B = 25.59.20$ nach (5)
 $E = 63.31.40$ nach (8)
 $C = 10.5.30$ nach (9)

2. Iteration: $D = 79.52$ nach (10)
 $B = 26.4.15$ nach (5)
 $E = 63.26.45$ nach (8)
 $C = 10.5$ nach (9), gegenüber
 $C = 10.5.30$ der 1. Iteration.

Die Kommutation berechnet sich nach (11):

$$A = 180 - (63.26.45 - 10.5)$$

$A = 126.38.15$ als Ausgangswert für die Berechnung von Statio prima und Statio secunda.

Statio prima, äußere Iteration:

$$v_{\odot} = 126.38.15 + 87.14.10 + 187.11.40 - 96.9.49$$

$$v_{\odot} = 304^{\circ} 54' 16'' \text{ nach (12)}$$

neue Ausgangswerte:

Sonne: $r = 1.101008 \quad \Delta M = 60' 36''$

Verhältnis der Tagbögen: $\frac{AB}{AD} = 2.25777$

Verhältnis der Abstände: $\frac{AC}{AE} = 5.14811.$

Statio prima, innere Iteration:

$$A = 180 - (63.39.55 - 10.1.38)$$

$$A = 126.21.43.$$

Wiederholung der Rechnung:

äußere Iteration:

$$v_{\odot} = 126.21.43 + 87.14.10 + 91.1.52$$

$$v_{\odot} = 304.37.45 \text{ gegenüber } 304.54$$

Sonne: $r = 1.01002 \quad \Delta M = 60' 36''$

$$\frac{AB}{AD} = 2.25777 \text{ wie vorher}$$

$$\frac{AC}{AE} = 5.14841$$

innere Iteration:

$$A = 180 - (63.39.28 - 10.1.28)$$

$$A = 126.22.$$

Wiederholung der Rechnung:

äußere Iteration:

$$v_{\odot} = 126.22 + 87.14.10 + 91.1.52$$

$$v_{\odot} = 304.38.2 \text{ gegenüber } 304.37.45$$

innere Iteration: Wiederholung entfällt.

Statio secunda, äußere Iteration:

$$\begin{aligned}v_{\odot} &= 87.14.10 - 126.38.15 + 91.1.52 \\v_{\odot} &= 51.37.47 \quad \text{nach (12)}\end{aligned}$$

$$\text{Sonne:} \quad r = 1.101097 \quad \Delta M = 60' 39''$$

$$\frac{AB}{AD} = 2.25591$$

$$\frac{AC}{AE} = 5.14357$$

innere Iteration:

$$\begin{aligned}A &= 180 - (63.37.53 - 10.1.54) \\A &= 126.24.1.\end{aligned}$$

Wiederholung der Rechnung:

äußere Iteration:

$$\begin{aligned}v_{\odot} &= 87.14.10 - 126.24.1 + 91.1.52 \\v_{\odot} &= 51.52.1 \text{ gegenüber } 51.37.47\end{aligned}$$

$$\text{Sonne:} \quad r = 1.01091 \quad \Delta M = 60' 39''$$

$$\frac{AB}{AD} = 2.25591 \text{ wie vorher}$$

$$\frac{AC}{AE} = 5.14388$$

innere Iteration:

$$\begin{aligned}A &= 180 - (63.37.53 - 10.1.51) \\A &= 126.23.58.\end{aligned}$$

Wiederholung der Rechnung:

äußere Iteration:

$$\begin{aligned}v_{\odot} &= 87.14.10 - 126.23.58 + 91.1.52 \\v_{\odot} &= 51.52.4 \text{ gegenüber } 51.52.1\end{aligned}$$

innere Iteration: Wiederholung entfällt.

Damit betragen die Kommutionen:

$$\begin{aligned}\text{für Statio prima} & \quad 126^{\circ} 22' \\ \text{für Statio secunda} & \quad 126^{\circ} 24'\end{aligned}$$

in genauer Übereinstimmung mit Keplers Werten (vgl. Tafel S. 5).

4. BESCHREIBUNG DES ALGOL-PROGRAMMS

Das ALGOL-Programm, das den Anweisungen Keplers bis in die Einzelheiten folgt, ist zur Berechnung der stationären Punkte in der scheinbaren Planetenbahn an der elektronischen Rechenanlage TR4 des Leibniz-Rechenzentrums der Bayerischen Akademie der Wissenschaften entwickelt worden.

Es benutzt bei der Lösung der Aufgabe nicht die astronomischen Tafeln von [1], sondern berechnet nach den Formeln von [2] die erforderlichen Größen Intervall, wahre Anomalie, Reduktion auf die Ekliptik usw.

Geringfügige Differenzen zu den Werten Keplers sind von vornherein zu erwarten, weil seine Tafelwerte gerundet sind und kleine Rechenfehler enthalten ([2], Abschnitt 1.21), während an der TR4 jeder Wert mit hoher Stellenzahl neu berechnet wird.

Das Programm verwendet die Programmiersprache ALGOL 60. Die Zeilen sind fortlaufend durchnummeriert von 1 ('BEGIN') bis 158 ('END',.).

Deutlich erkennbar ist die Blockstruktur (untergeordnete Blöcke) des Programms. Unterprogramme (Prozeduren) sind nicht verwendet worden (zur ALGOL-Sprache vgl. z. B. [4]). Um Programm und Ergebnisprotokoll übersichtlich zu halten, ist darauf verzichtet worden, durch eine übergeordnete Schleife die Anguli Commutationis für alle Planeten gleichzeitig zu berechnen.

Aus der Geometrie der stationären Punkte (Fig. 4) ist ersichtlich, daß bei einem unteren Planeten sich die Stellung Sonne-Planet im Vergleich zur Situation eines oberen Planeten umkehrt. Entsprechend kehren sich bei einem unteren Planeten auch die Verhältnisse der Entfernungen und Tagbögen um. Diese Unterscheidung nach oberen und unteren Planeten ist im Programm durch die Kennziffer K als ganzzahlige Größe 'INTEGER' (Zeile 3) berücksichtigt worden. Bei einem oberen Planeten wird für K 0, bei einem unteren Planeten 1 eingelesen (Zeile 8). Die Ausgangswerte werden zunächst für K = 0 berechnet, K = 1 wird in einer bedingten Anweisung als Sonderfall behandelt (Zeile 63).

Die Unterscheidung wird durchgehend beibehalten: in der inneren Iteration (Zeile 70 bis Zeile 84) und in der äußeren Iteration für Statio prima (Zeilen 93,102) und Statio secunda (Zeilen 132, 144).

Die *innere Iteration* (Zeile 70 bis Zeile 84) berechnet den Winkel C (im Programm WC bzw. WC1) und Winkel E (im Programm WE). Um die Wiederholung der Rechnung zu ermöglichen, wird eine Sprunganweisung (Zeile 83) verwendet, die auf eine vorangestellte Marke (Z. 71) springen läßt. Dazwischen liegt die Rechnung, die so oft durchlaufen wird, bis eine vorgegebene Genauigkeitsschranke für C (Zeile 80) erreicht ist, d. h. bis

$$\Delta C \rightarrow 1''.$$

Dann erst wird die Kommotation berechnet (Z. 85) und die *äußere Iteration* beginnt.

Ausgehend von der wahren Anomalie der Sonne, hat die äußere Iteration von vornherein die Unterscheidung Statio prima und Statio secunda zu treffen. Dieser Umstand ist im übergeordneten Block als Verteilervereinbarung STATIO mit den Marken PRIM und SEC berücksichtigt worden (Zeile 7). Die entsprechende Sprunganweisung (Z. 89) enthält zur Unterscheidung der Marken des Verteilers die echte Alternative [4] für eine weitere Kennziffer N. N ist am Anfang des Programms Null gesetzt worden (Z. 18). Da laut Anweisung (Z. 89) für N = 0 die Marke PRIM aufgerufen wird (Z. 90), berechnet das Programm zu-

```

1. 'BEGIN'
  !COMMENT! STATIONES DER PLANETEN NACH KEPLER..
  !INTEGER! EX, JAHR, K, L, N, P.,
  !REAL! AP, KN, A, E, I, M, APS, ES, MS, RO, V, W, B, R, T, EA, SIB,
5 WS, VS, EXS, RS, TS, VT, VR, WD, WC, WB, WE, WC1, ANCO, VS1.,
  !ARRAY! PLAN(/1..10/),
  !SWITCH! STATIO.=PRIM, SEC.,
  READ(K, PLAN, JAHR, AP, KN, A, E, I, M, APS, ES, MS),
  RO.=57.29577951.,
10 OUTPUT(1,
  !(*,/,/,!(TERMINI STATIONUM NACH KEPLERS TABULAE RUDOLPHINAE))',
  /,/!)',
  PRINT(PLAN),
  !FOR! EX.=0 !STEP! 30 !UNTIL! 330 !DO!
15 !BEGIN!
  OUTPUT(1, !(/,/,!(ANOMALIA ECCENTRI = !))',
  ZZD')', EX),
  L.=N.=P.=0.,
  EA.=EX/RO.,
20 !IF! EA 'EQUAL' 180/RO !THEN!
  V.=180/RO
  !ELSE!
  !BEGIN!
  !IF! EA 'LESS' 180/RO !THEN!
25 V.=2*ARCTAN(SQRT((1-E)/(1+E))*SIN(EA/2)/(COS(EA/2))),
  !ELSE!
  !BEGIN!
  V.=2*ARCTAN(SQRT((1-E)/(1+E))*SIN(180/RO-EA/2)/(COS(180/RO-EA/2))),
  V.=360/RO-V
30 !END!,
  !END!,
  W.=AP/RO+V.,
  !IF! W 'GREATER' 360/RO !THEN!
  W.=W-360/RO.,
35 SIB.=SIN(W-KN/RO)*SIN(I/RO),
  B.=ARCTAN(SIB/(SQRT(1-SIB*SIB))),
  R.=A*(1+E*COS(EA))*COS(B),
  T.=M/(1+E*SIN(EA)-E*SIN(EA-1/RO)),
  WS.=W-180/RO.,
40 !IF! WS 'LESS' 0 !THEN!
  WS.=360/RO+WS.,
  VS.=WS-APS/RO.,
  !IF! VS 'LESS' 0 !THEN!
  VS.=360/RO+VS.,
45 !COMMENT! BEGINN DER GROSSEN ITERATION.,
  ANFANG.,
  !IF! VS 'EQUAL' 180/RO !THEN! EXS.=180/RO
  !ELSE!
  !BEGIN!
50 !IF! VS 'GREATER' 180/RO !THEN!
  !BEGIN!
  EXS.=2*ARCTAN(SQRT((1+ES)/(1-ES))*SIN(180/RO-VS/2)/(COS(180/RO-VS/2))),
  EXS.=360/RO-EXS
  !END!
55 !ELSE!

```

BLATT 002

```


EXS.=2*ARCTAN(SQRT((1+ES)/(1-ES))*SIN(VS/2)/(COS(VS/2)))
'END'.,
RS.=1+ES*COS(EXS).,
TS.=MS/(1+ES*RO*SIN(EXS+1/RO)-ES*RO*SIN(EXS)).,
60 VT.=SQRT(A)*(1+ES*RO*SIN(EA+1/RO)-ES*RO*SIN(EA))/
(1+ES*RO*SIN(EXS+1/RO)-ES*RO*SIN(EXS)).,
VR.=R/RS.,
'IF' K 'EQUAL' 1 'THEN'
'BEGIN'
65 VT.=1/VT.,
VR.=1/VR
'END'.,
WD.=ARCTAN(1/VT/(SQRT(1-1/(VT*VT)))).,
WC.=89/RO-WD.,
70 'COMMENT' BEGINN DER INNEREN ITERATION.,
M1.,
'IF' K 'EQUAL' 0 'THEN'
WD.=90/RO-T/(2*RO)-WC
'ELSE' WD.=90/RO-TS/(2*RO)-WC.,
75 WB.=ARCTAN(SIN(WD)/VT/(SQRT(1-SIN(WD)/VT*SIN(WD)/VT))).,
'IF' K 'EQUAL' 0 'THEN'
WE.=90/RO-TS/(2*RO)-WB
'ELSE' WE.=90/RO-T/(2*RO)-WB.,
WC1.=ARCTAN(SIN(WE)/VR/(SQRT(1-SIN(WE)/VR*SIN(WE)/VR))).,
80 'IF' ABS(WC1-WC) 'GREATER' 1/(RO*60*60) 'THEN'
'BEGIN'
WC.=WC1.,
'GO TO' M1.
'END'.,
85 ANCO.=180/RO-WE+WC1.,
PRINT(L,P,VS*RO,ANCO*RO).,
'COMMENT' UNTERSCHIEDUNG DER STATIONES.,
'GO TO' Z.,
Z., 'GO TO' STATIO (/ 'IF' N 'EQUAL' 0 'THEN' 1 'ELSE' 2 /).,
90 PRIM.,
L.=1.,
P.=P+1.,
'IF' K 'EQUAL' 0 'THEN'
'BEGIN'
95 VS1.=ANCO+V+AP/RO-APS/RO.,
'IF' VS1 'NOT LESS' 360/RO 'THEN' VS1.=VS1-360/RO.,
'IF' ABS(VS1-VS) 'GREATER' 10/(RO*60*60) 'THEN'
'BEGIN'
VS.=VS1.,
100 'GO TO' ANFANG
'END'
'ELSE'
'BEGIN'
N.=1.,
105 P.=0.,
VS.=WS-APS/RO.,
'IF' VS 'LESS' 0 'THEN' VS.=360/RO+VS.,
'GO TO' Z
'END'.,
110 'END'
'ELSE'
'BEGIN'
VS1.=V-ANCO+AP/RO-APS/RO.,

```

BLATT 003

```
'IF' VS1 'LESS' 0 'THEN' VS1.=360/RO+VS1.,
115 'IF' ABS(VS1-VS) 'GREATER' 10/(RO*60*60) 'THEN'
'BEGIN'
VS.=VS1.,
'GO TO' ANFANG
'END'
120 'ELSE'
'BEGIN'
N.=1.,
P.=0.,
VS.=WS-APS/RO.,
125 'IF' VS 'LESS' 0 'THEN' VS.=360/RO+VS.,
'GO TO' Z
'END'.,
'END'.,
SEC.,
130 L.=2.,
P.=P+1.,
'IF' K 'EQUAL' 0 'THEN'
'BEGIN'
VS1.=V-ANCO+AP/RO-APS/RO.,
135 'IF' VS1 'LESS' 0 'THEN' VS1.=360/RO+VS1.,
'IF' ABS(VS1-VS) 'GREATER' 10/(RO*60*60) 'THEN'
'BEGIN'
VS.=VS1.,
'GO TO' ANFANG
140 'END'
'ELSE'
'GO TO' SCHLUSS.,
'END'
'ELSE'
145 'BEGIN'
VS1.=ANCO+V+AP/RO-APS/RO.,
'IF' VS1 'GREATER' 360/RO 'THEN' VS1.=VS1-360/RO.,
'IF' ABS(VS1-VS) 'GREATER' 10/(RO*60*60) 'THEN'
'BEGIN'
150 VS.=VS1.,
'GO TO' ANFANG
'END'
'ELSE'
'GO TO' SCHLUSS
155 'END'.,
SCHLUSS.,
'END'.,
'END'.,
```

STRUKTURDIAGRAMM ZUM ALGOL-PROGRAMM



Figur 7

nächst die Statio prima und die Statio secunda, wenn $N = 1$ gesetzt wird (Z. 104 für einen oberen, Z. 122 für einen unteren Planeten). Dabei erfolgt der Rückwärtssprung (Z. 108 bzw. Z. 126) auf die Marke Z, und die Marke SEC des Verteilers STATIO wird aufgerufen.

Die äußere Iteration läuft für beide Stationen gesondert und benutzt als Kriterium der Wiederholung eine Genauigkeitsschranke für die wahre Anomalie der Sonne (Z. 97, 115 bzw. Z. 136, 148). Die Rechnung wird wiederholt, bis

$$\Delta v_{\odot} \rightarrow 10''.$$

Die Rechnung verläuft so, daß nach einer äußeren Iteration durch eine Sprunganweisung (Z. 100, 118 bzw. Z. 139, 151) auf eine der inneren Iteration vorangestellte Marke (Z. 46) die Ausgangsdaten für die innere Iteration neu berechnet werden und dann die innere Iteration durchlaufen wird.

Ist die Schranke für v_0 erreicht, wird, nachdem beide Stationen berechnet sind, durch eine andere Sprunganweisung (Z. 142 bzw. Z. 154) der Vorwärtssprung auf eine Marke am Ende des Programms (Z. 156) ausgelöst.

Der hier beschriebene Programmteil läßt sich durch ein Strukturdiagramm (Fig. 7) verdeutlichen.

Nach den Kennziffern K (Unterscheidung in obere und innere Planeten) und N (Unterscheidung der Stationen) wird für das Ergebnisprotokoll eine weitere Kennziffer verwendet, um die Nummer der äußeren Iteration anzuzeigen: Kennziffer P. P wird in der Ausgangsrechnung Null gesetzt (Z. 18) und bei Iterationen jeweils um 1 erhöht (Z. 92, 131).

Die Ergebnisausgabe (wahre Anomalie der Sonne und Kommution) ist also durch 2 Ziffern gekennzeichnet: durch N und P.

N = 0 steht für den Ausgangswert (Opposition)

N = 1 steht für Statio prima

N = 2 steht für Statio secunda

P = 0 bezeichnet die Ausgangsdaten und gibt dann für

P = 1, 2, 3 die Nummer der äußeren Iteration bei jeder Stationsberechnung an.

Das gesamte Programm wird für die exzentrischen Anomalien von 0 bis 330° in Stufen von 30° für jeden Planeten durchlaufen (Z. 14). Die Ergebnisse werden für jede Anomalie sofort nach Berechnung der Kommution ausgedruckt (Z. 86).

Auf weitere Einzelheiten des Programms soll nicht weiter eingegangen werden, sie entsprechen dem vorangegangenen Rechenbeispiel und folgen formal den Regeln der ALGOL-Sprache [4].

5. DIE ERGEBNISSE DER NACHRECHNUNG KRITISCHER VERGLEICH MIT DEN WERTEN KEPLERS

Als Beispiel für den Ausdruck der Ergebnisse an der Rechenanlage TR4 wird das Ergebnisprotokoll für den Jupiter wiedergegeben.

Außerdem werden in einer Tabelle die Kommutionen aus der Rechnung an der TR4 den Keplerschen Werten aus den Tabellen Termini Stationum der einzelnen Planetentafeln aus [1] gegenübergestellt.

Die Kommutionen stimmen beim Jupiter am besten überein. Die maximale Differenz beträgt $1'$, sie resultiert aus der Rundung der Ergebnisse bei Kepler auf volle Minuten und entspricht den Erwartungen, wenn Kepler fehlerlos gerechnet hat.

Bei Saturn sind die Differenzen größer, Statio prima ist von Kepler genauer berechnet worden als Statio secunda. Für $E = 0$ läßt sich bei Kepler ein Rechenfehler von etwa $30'$ vermuten, nachweisen läßt er sich nicht, weil die entsprechenden Rechenunterlagen in den Handschriften fehlen.

Für Mars und Merkur, die Planeten mit größter Exzentrizität, sind zusätzlich für andere Anomalien Zwischenwerte von Kepler gerechnet worden, um besser interpolieren zu können.

TERMINI STATIONUM NACH KEPLERS TABULAE RUDDOLPHINAE

JUPITER

ANOMALIA ECCENTRI = 0			
0	0	2,7103083300. 2	1.2444161372. 2
1	1	2,1547244672. 2	1,2395873281. 2
1	2	2,1498956582. 2	1,2395593747. 2
1	3	2,1498677047. 2	1,2395592139. 2
2	1	3,2707491160. 2	1,2492900788. 2
2	2	3,2610182511. 2	1,2492341159. 2
2	3	3,2610742141. 2	1,2492344418. 2
ANOMALIA ECCENTRI = 30			
0	0	2,9967697693. 2	1,2493435956. 2
1	1	2,4461133649. 2	1,2438876078. 2
1	2	2,4406573771. 2	1,2438375845. 2
1	3	2,4406073538. 2	1,2438371270. 2
2	1	3,5529326422. 2	1,2523522131. 2
2	2	3,5444175560. 2	1,2523446495. 2
ANOMALIA ECCENTRI = 60			
0	0	3,2866565223. 2	1,2573345287. 2
1	1	2,7439910510. 2	1,2526126767. 2
1	2	2,7392691990. 2	1,2525632810. 2
1	3	2,7392198033. 2	1,2525627641. 2
2	1	2,3409375812. 1	1,2576923041. 2
2	2	2,2896421819. 1	1,2577144327. 2
ANOMALIA ECCENTRI = 90			
0	0	3,5826695872. 2	1,2665018276. 2
1	1	3,0491714149. 2	1,2638385001. 2
1	2	3,0465080873. 2	1,2638147053. 2
2	1	5,1885488190. 1	1,2640560696. 2
2	2	5,1861351760. 1	1,2640561574. 2
ANOMALIA ECCENTRI = 120			
0	0	2,8607914919. 1	1,2743555675. 2
1	1	3,3604347166. 2	1,2746245087. 2
1	2	3,3607036578. 2	1,2746257267. 2
2	1	8,1145342244. 1	1,2697064518. 2
2	2	8,1637269737. 1	1,2696528353. 2
2	3	8,1642631382. 1	1,2696522506. 2
ANOMALIA ECCENTRI = 150			
0	0	5,9619206000. 1	1,2784758772. 2
1	1	7,4667937261. 0	1,2816921644. 2
1	2	7,7884224449. 0	1,2816870034. 2
2	1	1,1145050565. 2	1,2728942403. 2
2	2	1,1232978196. 2	1,2728047291. 2
2	3	1,1233873307. 2	1,2728036210. 2
ANOMALIA ECCENTRI = 180			
0	0	9,1030833001. 1	1,2774818492. 2
1	1	3,8779017919. 1	1,2826921773. 2
1	2	3,9300050736. 1	1,2826540452. 2
1	3	3,9296237522. 1	1,2826543258. 2
2	1	1,4276540041. 2	1,2725684860. 2
2	2	1,4377398439. 2	1,2725032840. 2

206 T206-BIALAS A LRZ MUENCHEN 18.11.68 3

2	3	1.4378050459. 2	1.2725028676. 2
ANOMALIA ECCENTRI = 210			
0	0	1.2244246001. 2	1.2717084001. 2
1	1	6.9613350025. 1	1.2773343278. 2
1	2	7.0175892796. 1	1.2772746082. 2
1	3	7.0169920832. 1	1.2772752431. 2
2	1	1.7471493569. 2	1.2688843100. 2
2	2	1.7555402900. 2	1.2688772357. 2
ANOMALIA ECCENTRI = 240			
0	0	1.5345375110. 2	1.2630400147. 2
1	1	9.9757752569. 1	1.2674831015. 2
1	2	1.0020206125. 2	1.2674355393. 2
1	3	1.0019730502. 2	1.2674360479. 2
2	1	2.0671014630. 2	1.2630830248. 2
2	2	2.0714544861. 2	1.2631042128. 2
ANOMALIA ECCENTRI = 270			
0	0	1.8379470727. 2	1.2540392654. 2
1	1	1.2919863382. 2	1.2561859228. 2
1	2	1.2941329956. 2	1.2561686370. 2
2	1	2.3817784357. 2	1.2568774716. 2
2	2	2.3810696011. 2	1.2568711232. 2
ANOMALIA ECCENTRI = 300			
0	0	2.1339601377. 2	1.2470301439. 2
1	1	1.5809902816. 2	1.2464556342. 2
1	2	1.5804157720. 2	1.2464577717. 2
2	1	2.6875023660. 2	1.2518474578. 2
2	2	2.6821126799. 2	1.2517911397. 2
2	3	2.6821689979. 2	1.2517917280. 2
ANOMALIA ECCENTRI = 330			
0	0	2.4238468908. 2	1.2435892762. 2
1	1	1.8674361670. 2	1.2405169069. 2
1	2	1.8643637977. 2	1.2405132072. 2
2	1	2.9833336836. 2	1.2491235041. 2
2	2	2.9747233866. 2	1.2490441277. 2
2	3	2.9748027631. 2	1.2490448622. 2

64 SEC. THIS SECTION

ZUSAMMENSTELLUNG DER ERGEBNISSE

<i>Planet</i>	<i>E</i>	<i>STATIO PRIMA</i>		<i>STATIO SECUNDA</i>	
		<i>TR 4</i>	<i>Kepler</i>	<i>TR 4</i>	<i>Kepler</i>
SATURN	0	113° 47' 31''	113° 48'	113° 54' 53''	113° 57'
	90	115.26.42	115.27	114.40.47	114.47
	180	116.20.44	116.53	116.12.59	116.50
	270	114.35.54	114.37	115.22.21	115.24
JUPITER	0	123.57.21	123.57	124.55.24	124.54
	90	126.22.53	126.22	126.24.21	126.24
	180	128.15.55	128.15	127.15.01	127.15
	270	125.37.00	125.38	125.41.13	125.41
MARS	0	157.37.31	157.40	158.13.19	158.13
	60	160.52.55	160.22	161.30.00	161.13
	90	164.01.14	164.00	164.24.17	164.23
	120	167.15.11	167.30	167.20.51	167.17
	180	170.08.04	170.08	169.48.15	169.47
	240	165.55.30	165.08	165.27.02	165.08
	270	162.43.16	162.46	162.26.29	162.27
	300	159.55.10	159.56	159.53.57	159.54
VENUS	0	166.56.17	167.47	166.43.39	167.58
	90	167.59.50	167.58	167.35.47	167.35
	180	167.46.14	166.55	167.57.47	166.43
	270	166.41.26	166.43	167.07.03	167.07
MERKUR	0	154.58.02	153.48	155.17.24	154.10
	60	150.24.46	150.46	149.58.49	150.13
	90	145.54.10	144.58	145.05.45	145.09
	120	141.17.54	142.02	140.17.44	140.57
	180	136.18.59	136.46	135.56.06	136.23
	240	140.55.10	141.27	141.40.15	142.17
	270	145.42.48	145.19	146.39.55	146.28
	300	150.25.40	150.57	151.17.54	151.34

Die Hauptwerte ($E = 0, 90, 180, 270$) für Mars stimmen gut überein, Zwischenwerte haben z. T. erhebliche Differenzen bis zu 30'. Diese Ergebnisse deuten darauf hin, daß die Zwischenwerte nicht neu berechnet worden sind. Das Interpolationsverfahren läßt sich allerdings nicht sicher ermitteln.

Die Stationspunkte der Venus sind von Kepler wieder sehr genau berechnet worden (maximale Differenz 2'). Hier läßt sich in der Tabelle die Verwechslung zweier Werte nachweisen: die Kommationen von Aphel und Perihel sind entweder von Kepler oder vom Drucker vertauscht worden.

Die Ergebnisse der Merkur-Rechnung machen eine eingehende Diskussion erforderlich. In den Kepler-Handschriften ([3], Bd. 20, Bl. 28 u. ff.) sind Rechnungen zur scheinbaren Bahn des Merkurs erhalten. Diese Passagen sind nicht in das gedruckte Werk übernommen worden.

Kepler kommt zunächst allgemein auf die Berechnung der stationären Punkte zu sprechen und verweist auf Tabellen bei den einzelnen Tafeln. Er schreibt:

„Methodo jam tradita construxi tabellam in singulis planetis pro angulo isto, in cuius margine sinistro sunt numeri indicantes proportiones intervallorum temporaneorum...“

Diese Tabellen, die berechnet vorliegen, übernimmt Kepler nicht in das Werk, offensichtlich ist er später von dieser Berechnungsmethode wieder abgegangen.

Die Tafeln sind im ersten Teil von [1] in der „Tabula anguli potissimum pro prosthaphaeresibus orbis“ für alle Planeten zusammengefaßt.

Auf Bl. 29 in Bd. 20 von [3] berechnet Kepler die Statio prima von Merkur für die exzentrische Anomalie 0° . Die Zahlen entsprechen zunächst dem Rechenbeispiel in [1] für Merkur, S. 73. Für die wahre Anomalie der Sonne nimmt er 0° an. Offensichtlich liegen ihm bereits Näherungswerte älterer Tafelwerke vor.

Sein Verfahren, die Kommution zu berechnen, basiert zwar auf dem in [1] dargestellten Rechengang, weicht aber in der praktischen Ausführung von den Rechenbeispielen in [1] ab. Er nimmt nacheinander für den Winkel C (vgl. Fig. 4) verschiedene Ausgangswerte an, berechnet sich über die Verhältnisse der Intervalle und Tagbögen die Winkelsumme E, B, die er mit dem bekannten Wert für Summe E, B (Sollwert) vergleicht. Dieser Wert ergibt sich aus:

$$\Sigma E, B = 90^\circ - \frac{1}{2} d\varphi, \quad (13)$$

wobei $d\varphi$ der wahre Tagbogen des Merkurs ist und nach (2) berechnet wird. Das gesuchte C wird nach der Regula falsi interpoliert, dabei gilt als Kontrolle für die Richtigkeit der Rechnung:

$$\Sigma E, B \text{ (Istwert)} = \Sigma E, B \text{ (Sollwert)}.$$

Ohne die logarithmische Rechnung im einzelnen zu verfolgen, sollen die Ergebnisse der Winkelrechnungen und der Interpolationsrechnungen wiedergegeben werden, weil daraus der entscheidende Fehler Keplers in diesem Beispiel ersichtlich wird.

$$\text{Er setzt } C = 20^\circ \text{ und berechnet daraus } \begin{array}{l} E = 47^\circ 59' \\ B = 43.54 \end{array}$$

$$\begin{array}{rcl} \text{Istwert } \Sigma E, B & = & 91.53 \\ \text{Sollwert} & = & 88.18.30 \\ \hline D_1: \text{ Abweichung} & = & 3.34.30 \end{array}$$

$$\begin{array}{rcl} C = 19^\circ & & \begin{array}{l} E = 45.00 \\ B = 44.15 \end{array} \\ \text{Istwert } \Sigma E, B & = & 89.15. \end{array}$$

Für die Abweichung schreibt Kepler jetzt schon hier, ohne den Sollwert neu zu erwähnen:

$$D_2: \text{„excedit } 0.15 \frac{1}{2} \text{“}$$

$$\text{anstelle von } 0.56 \frac{1}{2}.$$

In der Rechnung taucht eine zweite Winkelsumme auf, die Kepler fälschlicherweise als Sollwert für $\Sigma E, B$ herangezogen hat:

$$\Sigma C, D = 90^\circ - \frac{1}{2} d_0 = 89.30 \frac{1}{2}.$$

In der Interpolationsrechnung erscheint zwar der richtige Wert für $D_1 - D_2 = 158'$, es wird aber für $D_2 = 0.15 \frac{1}{2}$, nicht für den richtigen Wert $0.56 \frac{1}{2}$, die Differenz in C ausgerechnet.

$$\text{Kepler rechnet: } C = C_2 - \frac{D_2}{D_1 - D_2} \cdot (C_1 - C_2) \quad (14)$$

$$= 19^\circ - \frac{(-15.30)}{3.34.30 - 56.30} (20 - 19)$$

$$= 19^\circ 6',$$

$$\text{anstelle von } C = 19^\circ - \frac{56.30}{3.34.30 - 56.30} (20 - 19)$$

$$= 18.38.30.$$

Ebenfalls nach der Regula falsi erhält Kepler für

$$E = 45^\circ 18',$$

$$\text{anstelle von } E = 43.56 \text{ mit } D_2 = 56' 30''.$$

Nach (11) berechnet Kepler die Kommotation zu

$$A = 180 - (45.18 - 19.6)$$

$$= 153^\circ 48' \text{ (der gedruckte Wert in [1])},$$

$$\text{anstelle von } A = 180 - (43.56 - 18.38.30)$$

$$= 154.42.30.$$

Dieser Wert stimmt mit dem Ergebnis der Nachrechnung an der TR4 (154.58) wesentlich besser überein.

In den Handschriften lassen sich andere Merkkurrechnungen finden, die ebenfalls nicht fehlerlos sind. Auf die Wiedergabe weiterer Beispiele soll verzichtet werden, weil sie für die Sache nichts Neues bringen.

Kepler, das sei abschließend zu diesem Punkt gesagt, ist sich mitunter eines Rechenfehlers bewußt, weiß aber, daß bei der schnellen Merkbewegung sich die Kommotation rasch ändert und ein Rechenfehler nicht allzu wichtig zu nehmen ist: „quod est nullius fere momenti“.

6. DIE LÖSUNG DES PROBLEMS ÜBER EINE DIFFERENTIALGLEICHUNG

Die von Kepler berechneten Kommutionen (A) für stationäre Stellungen der Planeten sind in [1] in den Tabellen Termini Stationum in Abhängigkeit von der exzentrischen Anomalie (E) angegeben.


Die graphische Darstellung der Funktion A (E) zeigt die in Fig. 8 wiedergegebenen Kurven für die einzelnen Planeten. Es handelt sich um Schwingungskurven symmetrisch zum Extremwert in $E = 180^\circ$. Die Amplitude zeigt eine Abhängigkeit von der numerischen Exzentrizität e, der Verlauf der Kurven deutet die Abhängigkeit von einer zweiten Größe an:

für obere Planeten gilt $\frac{dF}{dE} > 0$ für $0 < E < 180$

$\frac{dF}{dE} < 0$ für $180 < E < 360$

$\frac{dF}{dE} = 0$ für $E = 180$ (Maximum)

für $E = 0$ (Minimum).



Figur 8

Bei den unteren Planeten kehren sich die Vorzeichen der Tangenten um, das Minimum wird für $E = 180$ erreicht.

Aus der Kurvenbeschreibung folgt die Abhängigkeit von $\frac{1}{r}$, wobei r wiederum von e abhängt.

Die Kommutation wird sich also darstellen lassen als

$$A = F\left(\frac{1}{r}\right); \quad (15)$$

dieser Zusammenhang ist aus der Keplerschen Darstellung von Fig. 4 nicht zu ersehen.

Es ist zu erwarten, daß die Kurven von Fig. 8 Lösungen einer Differentialgleichung sind oder über eine Differentialgleichung zustande gekommen sind. Der Ansatz zur Differentialgleichung ergibt sich aus der Überlegung, daß die Sichtlinien Erde-Planet zu verschiedenen Zeitpunkten parallel sind, daß also die geozentrische Länge λ konstant bleibt:

$$\frac{d\lambda}{dt} = 0. \quad (16)$$

Diesen Gedanken hat J. Bauschinger in [5] bei der Betrachtung der scheinbaren Planetenbahn verfolgt (Abschnitt XVIII). Er geht vom Zusammenhang der geozentrischen Koordinaten λ, β, ρ eines Planeten mit den heliozentrischen Koordinaten der Erde und des Planeten aus. Die Grundgleichungen im Ekliptiksystem lauten:

$$\begin{aligned} \rho \cos \beta \cos (\lambda - \Omega) &= r \cos u + R \cos U \\ \rho \cos \beta \sin (\lambda - \Omega) &= r \sin u \cos i + R \sin u \\ \rho \sin \beta &= r \sin u \sin i. \end{aligned} \quad (17)$$

Es bedeuten: u Argument der Breite

i Bahnneigung gegen die Ekliptik

U Länge der Sonne, gezählt vom Knoten der Planetenbahn (Ω).

Wird von (17) das totale Differential gebildet, wobei die Bahnelemente konstant gehalten werden, ergibt sich nach [5] nach einigen Reduktionen und Umformungen die Gleichung:

$$\begin{aligned} \frac{\rho^2 \cos^2 \beta}{k} \frac{d\lambda}{dt} &= (r \sin u \cos i + R \sin U) \cdot \left\{ \frac{1}{\sqrt{p}} (\sin u - e \sin \omega) \right. \\ &\quad \left. + \frac{1}{\sqrt{p_0}} (\sin U + e_0 \sin \Omega) \right\} + (r \cos u + R \cos U) \cdot \\ &\quad \cdot \left\{ \frac{1}{\sqrt{p}} \cos i (\cos u - e \cos \omega) + \frac{1}{\sqrt{p_0}} (\cos U - e_0 \cos \Omega) \right\}. \end{aligned} \quad (18)$$

Hierin sind: p der Parameter der Bahnellipse des Planeten

p_0 der Parameter der Bahnellipse der Sonne

ω Argument des Planetenaphels

Ω Argument des Sonnenapogäums

k Gravitationskonstante.

In [5] vereinfacht Bauschinger die Zusammenhänge von (18) durch zwei Vernachlässigungen: er setzt zunächst $i = 0$, $\cos i = 1$, u wird jetzt in der Ekliptikebene gemessen.

Außerdem sieht er von der elliptischen Bahn ab und diskutiert (18) an dem Sonderfall der Kreisbahn.


Hier soll die erste Vernachlässigung beibehalten werden. Mit $i = 0$ und unter Beachtung von (16) vereinfacht sich (18) zu:

$$\cos (U - u) = A + B$$

mit
$$A = -\frac{p \sqrt{p_0} + p_0 \sqrt{p}}{r \sqrt{p} + R \sqrt{p_0}} \quad (19)$$

und
$$B = \frac{\sqrt{p \cdot p_0}}{r \sqrt{p} + R \sqrt{p_0}} \left\{ e \cos (U - \omega) + e_0 \cos (U - \Omega) \right\}.$$

Die Gleichung (19) stellt den Zusammenhang zwischen der Differenz Länge der Sonne — Länge des Planeten und den Bahnelementen her. Der Winkel $U - u$ ist aber nichts anderes als die gesuchte Kommution (Fig. 9).



Figur 9

Die Kommution (AC) stellt sich also dar als

$$AC = \pm \arccos (A + B). \quad (20)$$

Von (19) oder (20) läßt sich leicht auf die Kreisbahn schließen, indem gesetzt werden:

$$\begin{aligned} e &= e_0 = 0 \\ p &= r = a \quad (a \text{ ist die große Halbachse}) \\ p_0 &= R = 1. \end{aligned}$$

Die Gleichung für die Kommution der stationären Punkte einer Kreisbahn lautet:

$$\cos (U - u) = -\frac{1 + \sqrt{a}}{a + \frac{1}{\sqrt{a}}}. \quad (21)$$

Der Zusammenhang von (15) nach der Kurvenbetrachtung wird bestätigt, wenn in (21) Zähler und Nenner der rechten Seite durch a dividiert werden, a wieder durch r ersetzt

wird und in vereinfachender Betrachtung oberer Planeten die Größe $\frac{1}{r^{3/2}}$ vernachlässigt wird:

$$AC = \pm \arccos \left(\frac{1}{\sqrt{r}} + \frac{1}{r} \right).$$

Die Lösungen sind doppeldeutig (20), $\cos (U - u)$ bestimmt den Winkel $+(U - u)$. Bei oberen Planeten gilt das $+$ Zeichen für Statio prima, das $-$ Zeichen für Statio secunda, umgekehrt bei unteren Planeten.

Abschließend sei an einem Rechenbeispiel gezeigt, wie die Kommutation nach (20) und (21) berechnet wird. Die Auflösung der transzendenten Gleichung (20) nach der Kommutation ist durch Rechenwiederholung möglich.

Rechenbeispiel

Näherungslösung nach (20), für Saturn ($r = a$, $E = 90^\circ$):

$$a = 9.51, a = 3.08383$$

$$\cos (U - u) = -0.415265$$

Kommutation = $114^\circ 32'$ gegen $115^\circ 27'$ nach dem Verfahren von Kepler.

Rechnung nach (18) für Saturn ($E = 0$, Epoche 1. 1. 1626):

$$r = 10.05147$$

$$p = 9.47914 \text{ nach den Tafeln von [1]}$$

$$p_0 = 0.99968.$$

$$\begin{aligned} \text{Näherung: } \cos (U - u) &= -0.393015 \\ U - u &= 113^\circ 9'. \end{aligned}$$

Wiederholung: nach den Ausgangsdaten und Tafelwerten von [1]:

$$\begin{aligned} u &= 154.59 \\ U &= 268.08 \\ U - \omega &= 113.09, u - \Omega = 160.09, \end{aligned}$$

$$\begin{aligned} \cos (U - u) &= -0.392967 + 0.096347 \cdot (-0.022402 - 0.016948) \\ &= -0.396758 \end{aligned}$$

$$U - u = 113^\circ 22' \text{ gegen } 113^\circ 48' \text{ bei Kepler für Statio prima.}$$

Die Berechnung der Statio secunda verläuft analog, die Wiederholung der Rechnung geht von $-(U - u) = u - U$ aus.

Genaue Übereinstimmungen sind wegen der Vernachlässigung von i nicht zu erwarten.

7. SCHLUSSBETRACHTUNG

Die unscheinbaren Tabellen *TERMINI STATIONUM* in Keplers „Rudolphinischen Tafeln“ lassen kaum den großen Aufwand zu ihrer Berechnung vermuten. Erst das Studium des entsprechenden Textteils von [1] und die Durchmusterung des Druckmanuskripts in den Kepler-Handschriften ermöglichen die eingehende Diskussion der Tabellen und die wissenschaftliche Würdigung der Arbeiten Keplers.

Zum ersten Mal wird die scheinbare Planetenbahn im Lichte der neuen Bahntheorie Keplers gesehen. Was bei flüchtiger Betrachtung des Rechenganges wie eine einfache Interpolationsrechnung anmutet ([6], Nachbericht), erweist sich bei gründlicher Durchsicht als doppelte Iterationsrechnung, bei der in der inneren Iteration die Rechnung lediglich durch Interpolation nach der Regula falsi verkürzt wird.

Noch ein Wort zur Herausgabe der „Rudolphinischen Tafeln“ [6] durch die Kepler-Kommission der Bayerischen Akademie der Wissenschaften. Die Kommentierung des Werkes in der Form des Nachberichtes läßt die eingehende Diskussion einzelner Probleme nicht zu, eine derartige Diskussion soll durch die Neuherausgabe der Werke Keplers auch nicht „mitgeliefert“, sondern höchstens angeregt werden.

Mit der vorliegenden Arbeit wird nun der Anfang gemacht, die sowohl in den „Rudolphinischen Tafeln“ wie in dem handschriftlichen Nachlaß Keplers astronomischen Inhalts auftretenden Probleme in der Betrachtungsweise Keplers zu erörtern und die Rechnungen zur Lösung dieser Probleme mit modernen Hilfsmitteln nachzuvollziehen.

ZUSAMMENFASSUNG

Die stationären Punkte der scheinbaren Planetenbahn hat Kepler in unscheinbaren Tabellen für jeden Planeten angegeben. Die Durchsicht der „Rudolphinischen Tafeln“ und des Druckmanuskripts in den Handschriften Keplers läßt den Rechengang als zwei ineinander greifende Iterationsrechnungen erkennen. Die Nachrechnung der Tabellen über ein ALGOL-Programm an der elektronischen Rechanlage TR4 ergibt für den Planeten Merkur erhebliche Abweichungen, die sich in den Handschriften als Rechenfehler Keplers nachweisen lassen.

Die moderne Formulierung des Problems führt über eine einfache Differentialgleichung zu einer transzendenten Gleichung.

SUMMARY

For each planet Kepler gave the stationary points of the apparent planetary orbit in small inconspicuous tables. In scrutinizing the "Rudolphine Tables" and Kepler's manuscripts one can recognize his mathematical procedure as consisting of two interlocking iteration calculations.

Checking the tables with a computer TR4 (ALGOL-program) reveals considerable deviations for Mercury, which result from the miscalculations in Kepler's manuscripts.

The modern formulation of the problem leads over a simple differential equation to a transcendental equation.

LITERATURNACHWEIS

- [1] Kepler, Johannes: *Tabulae Rudolphinae*. Ulm 1627.
- [2] Bialas, Volker: *Die Rudolphinischen Tafeln von Johannes Kepler – Mathematische und astronomische Grundlagen*. Nova Kepleriana N. F. Heft 2 (Abhandl. d. Bayer. Akad. d. Wissenschaften, Mathematisch-Naturwiss. Klasse, N. F. Heft 139). München 1969.
- [3] Pulkowoer Kepler-Manuskripte.
- [4] Müller, Dieter: *Programmierung elektronischer Rechenanlagen*. B-I-Hochschultaschenbuch. Mannheim 1966.
- [5] Bauschinger, Julius: *Die Bahnbestimmung der Himmelskörper*. Leipzig 1906.
- [6] Kepler, Johannes: *Tabulae Rudolphinae*. Johannes Kepler, *Gesammelte Werke* Bd. X. München 1969.